
The Model

!variables:transition

LGDP, LGDP_BAR, G, Y, GROWTH4, GROWTH, GROWTH4_BAR, GROWTH_BAR
PIE4, PIELTE
UNR, UNR_BAR, UNR_GAP, UNR_G
CAPU, CAPU_BAR, CU_G, CAPU_GAP,
’Population’ LN, ’Population Growth’ DLN,
’Log Participation Rate’ LP, ’Log Participation Rate Trend’ LP_BAR,
’Log Labor Force Trend’ LLF_BAR, ’Log Labor Force’ LLF,
’Labor Force Trend Growth’ DLLF_BAR, ’Labor Force Growth’ DLLF,
’Log Employment’ LE, ’Employment Growth’ DLE, ’Y/Y Employment Growth’ DLE4,
’Log Employment Trend’ LE_BAR, ’Employment Trend Growth’ DLE_BAR,
’Y/Y Employment Trend Growth’ DLE4_BAR,
’Avg Labor Productivity’ APL, ’Avg Labor Productivity Growth’ DAPL,
’Y/Y Avg Labor Productivity Growth’ DAPL4,
’Avg Labor Productivity Trend’ APL_BAR, ’Avg Labor Productivity Trend Growth’ DAPL_BAR,
’Y/Y Avg Labor Productivity Trend Growth’ DAPL4_BAR,
’Participation gap’ LP_GAP

!shocks:transition

RES_APL_BAR, RES_G, RES_Y
RES_PIE, RES_PIELTE
RES_UNR_BAR, RES_UNR_G, RES_UNR_GAP
RES_CAPU_BAR, RES_CU_G, RES_CAPU_GAP
’Population Growth Shock’ RES_DLN
’Participation Rate Gap Shock’ RES_LP
’Participation Rate Trend Shock’ RES_LP_BAR

!parameters

lab_share, tau, growth_ss, rho1, rho2
beta, omega
parhist, lambda, unr_ss, alpha, phi1, phi2
delta, kappa1, kappa2
’Steady-State Population Growth’ popgrowth_ss,
’Steady-State Participation Rate’ part_ss

’Population Growth Dynamics’ zeta
’Discouraging Effect on Participation’ theta2
’Participate Rate Dynamics’ eta
theta1

!equations:transition

% Output.
LGDP = LGDP_BAR + Y;
Y = rho1*Y{-1} - rho2/100*(PIE4{-1} - PIELTE{-1}) + RES_Y;
GROWTH4 = LGDP - LGDP{-4};
GROWTH = 4*(LGDP - LGDP{-1});
GROWTH4_BAR = LGDP_BAR - LGDP_BAR{-4};
GROWTH_BAR = 4*(LGDP_BAR - LGDP_BAR{-1});

% Inflation.
PIE4 = PIE4{-1} + beta*Y + omega*(Y - Y{-1}) + RES_PIE;
PIELTE = PIELTE{-1} + RES_PIELTE;

% Unemployment.
UNR = UNR_BAR - UNR_GAP;
UNR_BAR = UNR_BAR{-1} + UNR_G - parhist/100*Y{-1} - lambda/100*(UNR_BAR{-1} - unr_ss) + RES_UNR_BAR;
UNR_G = (1-alpha)*UNR_G{-1} + RES_UNR_G;
UNR_GAP = phi1*UNR_GAP{-1} + phi2*Y + RES_UNR_GAP;

% Capacity utilization.
CAPU = CAPU_BAR + CAPU_GAP;
CAPU_BAR = CAPU_BAR{-1} + CU_G + RES_CAPU_BAR;
CU_G = (1 - delta)*CU_G{-1} + RES_CU_G;
CAPU_GAP = kappa1*CAPU_GAP{-1} + kappa2*Y + RES_CAPU_GAP;

% Average productivity of labor.
APL_BAR = APL_BAR{-1} + G/4 + RES_APL_BAR;
G = (1-tau)*G{-1} + tau*growth_ss + RES_G;
DAPL_BAR = 4*(APL_BAR - APL_BAR{-1});
DAPL4_BAR = APL_BAR - APL_BAR{-4};
DAPL = 4*(APL - APL{-1});
DAPL4 = APL - APL{-4};

% Potential output is residually determined from this equation.

APL_BAR = LGDP_BAR - LE_BAR;
APL = LGDP - LE;

% Population.
DLN = (1-zeta)*DLN{-1} + zeta*popgrowth_ss + RES_DLN;
DLN = 4*(LN - LN{-1});

% Labor market participation.
LP_GAP = theta1*LP_GAP{-1} -100/part_ss*theta2*(UNR - UNR_BAR) + RES_LP;
LP_GAP = LP - LP_BAR;
LP_BAR = (1-eta)*LP_BAR{-1} + eta*100*log(part_ss/100) + RES_LP_BAR;

% Labor force trend.
LLF_BAR = LN + LP_BAR;
DLLF_BAR = 4*(LLF_BAR - LLF_BAR{-1});
LLF = LN + LP;
DLLF = 4*(LLF - LLF{-1});

% Employment trend.
% Mixed first-order expansion of UNR_BAR = 1 - E_BAR/LF_BAR;
UNR_BAR - unr_ss = -(1-unr_ss/100)*(LE_BAR - LLF_BAR - 100*log(1-unr_ss/100));
DLE_BAR = 4*(LE_BAR - LE_BAR{-1});
DLE4_BAR = LE_BAR - LE_BAR{-4};
UNR - unr_ss = -(1-unr_ss/100)*(LE - LLF - 100*log(1-unr_ss/100));
DLE = 4*(LE - LE{-1});
DLE4 = LE - LE{-4};

!variables:measurement

GROWTH_, PIE4_, D_CAPU_, UNR_, PIE4_PIELTE_
’Log Labour Force’ LLF_, ’Log Participation Rate’ LP_

%These variables are added for tuning
LP_BAR_
DAPL_BAR_

!shocks:measurement

RES_LLF_
RES_DAPL_BAR_

!equations:measurement

GROWTH_ = GROWTH;
PIE4_ = PIE4;
D_CAPU_ = CAPU - CAPU{-1};
UNR_ = UNR;
PIE4_PIELTE_ = PIE4 - PIELTE;
LLF_ = LLF + RES_LLF_;
LP_ = LP;

%These equations are added for tuning
LP_BAR_ = LP_BAR;
DAPL_BAR_ = DAPL_BAR + RES_DAPL_BAR_;

Economic activity gaps

Output Gap +/- 2 std devs (w/o model or parameter uncertainty)

Output Growth (Q/Q Annualized) +/- 2 std devs (w/o model or parameter uncertainty)

Output Growth (Y/Y) +/- 2 std devs (w/o model or parameter uncertainty)

Growth of Labor Productivity (Q/Q Annualized)

Growth of Labor Productivity (Y/Y)

Labor Productivity and Output Trend Growth (Q/Q Annualized)

Unemployment Rate

Growth of Employment (Q/Q Annualized)

Growth of Employment (Y/Y Annualized)

Participation Rate (in percent)

Labor Force And Population Growth (Q/Q Annualized)

Equilibrium Capacity Utilization +/- 2 std devs (w/o model or parameter uncertainty)

Relative Variances of Potential Growth and Changes in the Output Gap at Different Horizons

Regularized Maximum Likelihood

Parameter Prior Posterior
Mode Dispersion Mode Dispersion

alpha 0.500 0.158 0.496 0.021
beta 0.400 0.316 0.156 0.026
omega 0.500 0.316 0.251 0.036
rho1 0.800 0.158 0.822 0.018
kappa1 0.100 0.632 0.435 0.033
phi1 0.800 0.158 0.723 0.018
phi2 0.300 0.158 0.240 0.014
tau 0.100 0.158 0.117 0.020
delta 0.500 0.158 0.495 0.021
kappa2 1.500 1.581 1.222 0.060
parhist 5.000 3.162 4.898 0.375
rho2 5.000 3.162 5.082 0.417
lambda 1.000 3.162 1.045 0.226
zeta 0.100 0.079 0.082 0.010
eta 0.020 0.016 0.011 0.002
theta2 0.100 0.079 0.064 0.008
theta1 0.900 0.079 0.898 0.010
std RES Y 1.000 0.316 0.788 0.037
std RES G 1.000 0.316 0.753 0.043
std RES UNR GAP 0.500 0.316 0.221 0.023
std RES UNR BAR 0.020 0.016 0.017 0.002
std RES UNR G 0.020 0.016 0.017 0.002
std RES CAPU GAP 0.400 0.316 0.407 0.033
std RES CAPU BAR 0.250 0.158 0.241 0.020
std RES CU G 0.075 0.032 0.063 0.004
std RES PIE 0.500 0.316 0.516 0.029
std RES PIELTE 0.300 0.316 0.076 0.009
std RES APL BAR 0.250 0.079 0.189 0.010
std RES DLN 0.150 0.119 0.143 0.013
std RES LP 0.150 0.119 0.171 0.012
std RES LP BAR 0.100 0.079 0.136 0.008

Calibrated Parameters

Parameter Value
lab share 0.700
growth ss 1.500
unr ss 5.450
popgrowth ss 1.000
part ss 64.000
std RES LLF 0.821
std RES DAPL BAR 0.930

RMSEs

Parameter 1Q Ahead 4Q Ahead 8Q Ahead 12Q Ahead
GROWTH 3.412 3.363 3.370 3.414
PIE4 0.515 1.404 1.998 2.369
UNR 0.300 1.020 1.607 1.779

Mean absolute revisions (according to most recent estimates)

quarter t−12 t−8 t−4 t (nowcast)
Y 0.051 0.090 0.156 0.231
Y (HP) 0.172 0.333 0.754 1.250
UNR GAP 0.051 0.047 0.050 0.067
UNR GAP (HP) 0.086 0.176 0.393 0.640

Real-time output gap (MV)

Real-time output gap (HP)

